

**Panel Study of Income Dynamics 1999
Family Food Security Status Data File
Technical Documentation and User Notes**

Data File and Documentation Prepared by
Mark Nord
Economic Research Service
United States Department of Agriculture
April 29, 2002

Revision history of this document:

April 29, 2002 - original creation of document

Background

The Panel Study of Income Dynamics 1999 Family Food Security Status Data File (hereafter Food Security Status Data File) contains summary food security status information for families in the PSID 1999 Family Data File. The summary food security status variables were calculated from responses to the 18 questions in the U.S. Food Security Survey Module, which were collected in the regular 1999 interviews of PSID families. Two sets of food security variables are included: one set is based on the Household Food Security Scale, and the other (for families with children) is based on the Children's Food Security Scale.

The Food Security Status Data File contains only the food security summary variables and the Family Interview ID Number. It is intended to be matched to, and used in conjunction with, the PSID 1999 Family Data File and other PSID files to which the PSID 1999 Family Data File can be matched. The PSID data files contain extensive information about families and individuals in 1999 and information about some individuals and their families for earlier years, in some cases as far back as 1968. These data files, along with documentation and other information on the study are available from the PSID Main Data Center at <http://stat0.isr.umich.edu/psid/data-center/dcmain.html>.

Technical Description

The Food Security Status File is an ASCII file containing 6,997 records. The length of each record is 23 characters. The file is also available as a SAS-PC data file. The data dictionary is included below as Appendix A. SAS code to read the ASCII data file and create a SAS file is included as Appendix B. SAS code to create the food security summary variables directly from variables in the PSID 1999 Family Data File (obviating the need to match to the Food Security Status Data File) is included as Appendix C. Frequency tabulations of each of the six food security status variables are included as Appendix D.

Matching to the PSID 1999 Family Data File

The Food Security Status File contains one record for each family in the PSID 1999 Family Data File. It matches to that data file by ER13002 (1999 FAMILY INTERVIEW (ID) NUMBER). Both the Food Security Status Data File and the PSID 1999 Family Data File are sorted in this order. Information on matching to other PSID files is available from the PSID web site listed above.

Cases with No Valid Responses to Food Security Status Variables

For 3 families, the respondent provided no valid answer to any of the first three question in the food security scale, and was screened out of the remaining questions in the module. The food security status of these families is unknown, and the food security status variables for them are coded missing (-9).

Food Security Status Variables

Food security status variables were calculated based on the 18 core items in the food security module, ER14308 through ER14331 (excluding ER14311, 14322, and 14326, which specify screener calculations

and whether there were children in the family). Calculations were carried out in accordance with the standard methods described in *Measuring Household Food Security* (Bickel et al, 2000, available from the U.S. Department of Agriculture's Food and Nutrition Service Web Site, <http://www.fns.usda.gov/oane/>) and *Measuring the Food Security of Children in U.S. Households, 1995-1999* (Nord and Bickel, 2002, available from the U.S. Department of Agriculture's Economic Research Service Web Site, <http://www.ers.usda.gov/briefing/foodsecurity/>). ERS assessment of PSID 1999 responses to the food security questions using statistical methods based on the Rasch measurement model indicated that the use of the standard methods and scores for the food security status variables was appropriate.

No imputation was carried out for missing responses. Excluding the 3 families that gave no valid responses, and considering as valid those responses to questions that were skipped because of screening, only 9 families (0.13 percent of those interviewed) had any items missing. Effectively, their missing responses were treated as negatives, that is, as not indicating a food access problem.

Six food security status variables are provided as follows:

FSRAW is the household food security raw score, a simple count of the number of food security items affirmed by the family respondent. This ranges from 0 to 18 for families with children and 0 to 10 for families without children.

FSSCAL is the household (or family) food security scale score. This is a measure of the severity of food insecurity or hunger experienced in the family during calendar year 1998. It is a continuous, interval-level measure based on the Rasch measurement model and is appropriate for linear models such as correlation, regression, or analysis of variance. It is on the standard computational metric described in *Measuring Household Food Security*. Valid values range from 1.43 to 13.03, with higher values indicating more severe food deprivation. Technically, the scale score is undefined for families that affirmed no items. These families were food secure, but the appropriate size of the interval between their score and the score of families that affirmed one item is not known and varies from family to family. The variable is coded -6 for families that affirmed no items (or were screened out and deemed to be food secure) to remind users that these cases require special consideration in analyses.

FSSTAT is a categorical measure of food security status that identifies families as food secure, food insecure without hunger, and food insecure with hunger. This variable is appropriate for comparing prevalence rates of food insecurity and hunger across subpopulations.

FSCHRAW is the children's food security raw score, a count of the number of child-referenced food security items affirmed by the family respondent. This ranges from 0 to 8 for families with children and is coded -1 (not in universe) for families without children. Families with ER14322 (F38 CKPT: WTR CHILD < 18 in FU LAST YR) = 5 were not administered any of the child-referenced items and were assumed to have had no children in the family. Families with ER14322 = 0 had been screened out prior to ER14322 and had responded "never" to the first three household food security scale questions. For these families, presence of children was assessed by ER13013 (# CHILDREN IN FU). Families with ER13013 = 0 were assumed to have had no children in the family. Families with ER13013 > 0 were assumed to have had children in the family and assigned FSCHRAW=0.

FSCHSCAL is the children's food security scale score. This is a measure of the severity of food insecurity or hunger experienced by children in the family during calendar year 1998. Like the corresponding household food security scale score, it is a continuous, interval-level measure based on the Rasch measurement model and is appropriate for linear models such as correlation, regression, or analysis of variance. It is on the same metric as the household food security scale score. Valid values range from 4.11 to 12.25 with higher values indicating more severe food deprivation. The scale score is undefined for families that affirmed no child-referenced items. These families registered no food access problems that

affected children's eating patterns or food intake, but the appropriate size of the interval between their score and the score of families that affirmed one child-referenced item is not known and may vary from family to family. The variable is coded -6 for families that affirmed no child-referenced items (or were screened out and deemed to be food secure) to remind users that these cases require special consideration in analyses.

FSCHSTAT is a categorical measure of children's food security status that identifies families with hunger among children (one or more children in the household). This variable is appropriate for comparing prevalence rates of hunger among children across subpopulations.

Appendix A.
Data Dictionary for PSID 1999 Family Food Security Status Data File

 * FAMILY MATCH VARIABLE *

DATA SIZE BEGIN
 D ER13002 5 1
 T 1999 Family Interview (ID) Number
 U All families
 V 1:10071 .number

 * FOOD SECURITY STATUS VARIABLES *

D FSRAW 2 6
 T Household Food Security Raw Score
 T (number of affirmative responses)
 U All families
 V -9 .Missing, no valid responses
 V 0 .No affirmative responses
 V 1:18 .Number of affirmative responses

D FSSCAL 5 8
 T Household Food Security Scale Score (Rasch score)
 T (2 decimal places, decimal point in ASCII file)
 U All families
 V -9 .Missing, no valid responses
 V -6 .No affirmative responses
 V .scale score undefined
 V 1.43:
 V 13.03 .Number

D FSSTAT 2 13
 T Household Food security Status Category
 U All families
 V -9 .Missing, no valid responses
 V 1 .Food secure
 V 2 .Food insecure without hunger
 V 3 .Food insecure with hunger

D FSCHRAW 2 15
 T Children's Food Security Raw Score
 T (number of affirmative responses)
 U Families with children < 18
 V -9 .Missing, no valid responses
 V -1 .Not in universe, no child present
 V 0 .No affirmative responses
 V 1:8 .Number of affirmative responses

D FSCHSCAL 5 17
T Children's Food Security Scale Score (Rasch score)
T (2 decimal places, decimal point in ASCII file)
U Families with children < 18
V -9 .Missing, no valid responses
V -6 .No affirmative responses
V .scale score undefined
V -1 .Not in universe, no child present
V 4.11:
V 12.25 .Number

D FSCHSTAT 2 22
T Children's Food Security Status Category
U Families with children < 18
V -9 .Missing, no valid responses
V -1 .Not in universe, no child present
V 1 .No hunger among children
V 2 .Food insecure with hunger among children

Appendix B

SAS Program to Read PSID 1999 Food Security Status ASCII Data File

```
*****create SAS file from ascii food security file;
data temp;
infile 'd:\psid99fs.dat' lrecl=23; *specify the correct path to psid99fs.dat on your computer;
input
@1 er13002      5.
@6 fsraw       2.
@8 fsscal      f5.2
@13 fsstat     2.
@15 fschraw    2.
@17 fschscal   f5.2
@22 fschstat   2.;
run;
```

*This file can now be merged to the PSID 1999 Family Data File (Public Release I) by er13002;
*Both files are already sorted in this order;

Appendix C.
SAS Program to Create Food Security Status Variables Directly from
Variables in the PSID Family Data File
(Making it Unnecessary to Match to the Food Security Data File)

*This programming assumes that the food security scale variables and associated screening variables (ER14308 through ER14331) and ER13013 (# children in fu) are in the source file (psidfam), with variable names as in the PSID 1999 Family Data File (Public Release I);
 *The food security status variables created by this program have the same names and meanings as in the data dictionary in Appendix A above;

```
data psidfs (drop=hch1--hch8 q); set psidfam;
***specify appropriate sas dataset names for your computing environment;
array qfsec {18} er14308 er14309 er14310 er14323 er14324 er14325 er14312 er14313 er14314
  er14315 er14316 er14317 er14318 er14327 er14328 er14329 er14330 er14331;
*items in std order as in CPS Food Security Supplement 1998 and later, not in PSID order;
*
  worried fnotlast balmeal chfewfd chbal chenuf cutskip cutskipf eatless
  hungry losewt whlday whldayf chcut chskip chskipf chhungry chwhlday;
if _n_ eq 1 then do; *set lookup tables for hh scores;
  hch1=1.43; hch2=2.56; hch3=3.40; hch4=4.14; hch5=4.81; hch6=5.43; hch7=6.02; hch8=6.61;
  hch9=7.18; hch10=7.74; hch11=8.28; hch12=8.79; hch13=9.31; hch14=9.84; hch15=10.42;
  hch16=11.13; hch17=12.16; hch18=13.03;
  hnoch1=1.72; hnoch2=3.10; hnoch3=4.23; hnoch4=5.23; hnoch5=6.16;
  hnoch6=7.07; hnoch7=8.00; hnoch8=8.98; hnoch9=10.15; hnoch10=11.05;
  hchch1=4.11; hchch2=5.85; hchch3=7.54; hchch4=8.79;
  hchch5=9.62; hchch6=10.45; hchch7=11.50; hchch8=12.25;
  *standard scores, std computational metric, from 1998 rounded to 2 decimals;
  array hchsc{18}hch1--hch18;
  array hnochsc{10}hnoch1--hnoch10;
  array hchchsc{8}hchch1--hchch8;
end; *of set lookup tables for hh scores;
retain hch1--hch8;
*calculate raw scores for hh scale and child scale;
fsraw=0; fschraw=0;
do q=1 to 18; *add contribution of each item to raw score;
  if q in (1,2,3,4,5,6,8,13,16) then do; *often/sometimes/never or how often item;
    if qfsec{q} in (1,2) then fsraw=fsraw+1;
    if q in (4,5,6,16) and qfsec{q} in (1,2) then fschraw=fschraw+1;
  end; *of often/sometimes/never or how often item;
  else if q in (7,9,10,11,12,14,15,17,18) then do; *yes/no item;
    if qfsec{q} eq 1 then fsraw=fsraw+1;
    if q in (14,15,17,18) and qfsec{q} eq 1 then fschraw=fschraw+1;
  end; *of often/sometimes/never or how often item;
end; *of add contribution of each item to raw score;
if er14308 gt 3 and er14309 gt 3 and er14310 gt 3 then do; *no valid response;
  fsraw=-9;
  fschraw=-9;
end; *no valid response;
if er14322 eq 5 or (er14322 eq 0 and er13013 eq 0) then fschraw=-1; *no child in family;
*calculate food security scale scores for hh scale and child scale;
if fsraw eq -9 then do; fsscal=-9; fschscal=-9; end; *no valid fs response;
if fschraw eq -1 then fschscal=-1; *no child in family;
if fsraw ne -9 then do; *valid fs response;
  if er14322 eq 5 or (er14322 eq 0 and er13013 eq 0) then do; *no child in family;
    if fsraw eq 0 then fsscal=-6; *no affirmative responses;
    else fsscal=hnochsc{fsraw};
    fschscal=-1;
  end; *of no child in family;
else do; *child in family;
```

```

    if fsraw eq 0 then fsscal=-6; *no affirmative responses;
    else fsscal=hchsc{fsraw};
    if fschraw eq 0 then fschscal=-6; *no affirmative response to child scale;
    else fschscal=hchchsc{fschraw};
    end; *of child in family;
end; *of valid fs response;
*assign food security status categories;
if fsscal eq -9 then fsstat=-9; *no valid response;
else if fsscal gt 6.38 then fsstat=3; *insecure with hunger;
else if fsscal gt 3.25 then fsstat=2; *insecure without hunger;
else fsstat=1; *food secure;
if fschscal eq -9 then fschstat=-9; *no valid response;
else if fschscal eq -1 then fschstat=-1; *no child in family;
else if fschscal gt 9.14 then fschstat=2; *hunger among children;
else fschstat=1; *no hunger among children;

proc freq data=psidfs;
tables fsraw fsscal fsstat fschraw fschscal fschstat;
title1 'PSID family records, unweighted';
run;

```

Appendix D.
Frequency Distributions of Variables in PSID 1999 Food Security Status Data File

The **FREQ** Procedure

fsraw	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	3	0.04	3	0.04
0	5679	81.16	5682	81.21
1	343	4.90	6025	86.11
2	267	3.82	6292	89.92
3	207	2.96	6499	92.88
4	121	1.73	6620	94.61
5	95	1.36	6715	95.97
6	79	1.13	6794	97.10
7	57	0.81	6851	97.91
8	49	0.70	6900	98.61
9	25	0.36	6925	98.97
10	30	0.43	6955	99.40
11	11	0.16	6966	99.56
12	8	0.11	6974	99.67
13	12	0.17	6986	99.84
14	1	0.01	6987	99.86
15	5	0.07	6992	99.93
16	3	0.04	6995	99.97
17	1	0.01	6996	99.99
18	1	0.01	6997	100.00

fsscal	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	3	0.04	3	0.04
-6	5679	81.16	5682	81.21
1.43	212	3.03	5894	84.24
1.72	131	1.87	6025	86.11
2.56	171	2.44	6196	88.55
3.1	96	1.37	6292	89.92
3.4	123	1.76	6415	91.68
4.14	71	1.01	6486	92.70
4.23	84	1.20	6570	93.90
4.81	64	0.91	6634	94.81
5.23	50	0.71	6684	95.53
5.43	53	0.76	6737	96.28
6.02	34	0.49	6771	96.77
6.16	31	0.44	6802	97.21
6.61	29	0.41	6831	97.63
7.07	26	0.37	6857	98.00
7.18	19	0.27	6876	98.27
7.74	18	0.26	6894	98.53
8	23	0.33	6917	98.86
8.28	11	0.16	6928	99.01
8.79	8	0.11	6936	99.13
8.98	20	0.29	6956	99.41
9.31	12	0.17	6968	99.59
9.84	1	0.01	6969	99.60
10.15	6	0.09	6975	99.69

10.42	5	0.07	6980	99.76
11.05	12	0.17	6992	99.93
11.13	3	0.04	6995	99.97
12.16	1	0.01	6996	99.99
13.03	1	0.01	6997	100.00

fsstat	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	3	0.04	3	0.04
1	6289	89.88	6292	89.92
2	510	7.29	6802	97.21
3	195	2.79	6997	100.00

fschraw	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	1	0.01	1	0.01
-1	3575	51.09	3576	51.11
0	3043	43.49	6619	94.60
1	176	2.52	6795	97.11
2	104	1.49	6899	98.60
3	44	0.63	6943	99.23
4	29	0.41	6972	99.64
5	12	0.17	6984	99.81
6	5	0.07	6989	99.89
7	5	0.07	6994	99.96
8	3	0.04	6997	100.00

fschscal	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	1	0.01	1	0.01
-6	3043	43.49	3044	43.50
-1	3575	51.09	6619	94.60
4.11	176	2.52	6795	97.11
5.85	104	1.49	6899	98.60
7.54	44	0.63	6943	99.23
8.79	29	0.41	6972	99.64
9.62	12	0.17	6984	99.81
10.45	5	0.07	6989	99.89
11.5	5	0.07	6994	99.96
12.25	3	0.04	6997	100.00

fschstat	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	1	0.01	1	0.01
-1	3575	51.09	3576	51.11
1	3396	48.54	6972	99.64
2	25	0.36	6997	100.00